# $Id: PKGBUILD 86772 2013-03-22 00:29:34Z eric $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: Aaron Schaefer <aaron@elasticdog.com>
# Contributor: Erwin Van de Velde <erwin.vandevelde@gmail.com>

pkgname=rdiff-backup
pkgver=1.2.8
pkgrel=5
pkgdesc='A utility for local/remote mirroring and incremental backups.'
arch=('i686' 'x86_64')
url='http://www.nongnu.org/rdiff-backup/'
license=('GPL')
depends=('python2' 'librsync')
optdepends=('pylibacl: access control list support'
            'pyxattr: extended attribute support')
source=("http://savannah.nongnu.org/download/${pkgname}/${pkgname}-${pkgver}.tar.gz"{,.asc})
md5sums=('1a94dc537fcf74d6a3a80bd27808e77b'
         '3b88b18eb43ac85bc70b8dce92e958ba')

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  python2 setup.py install --root="${pkgdir}" -O1

  # Fix for FS#18848
  sed -i 's|python2|python2 -W ignore::DeprecationWarning|' "${pkgdir}/usr/bin/rdiff-backup"
}
